
Objectives 
 

Program Purpose 
 
• To generate a random pattern of colours, store 

the pattern and recall it. 
• To alter the shape style of the pattern, store 

and recall the shapes. 
 

Learning Goals 
 
• Further use of Control and Variable Arrays 
• Understanding of various shape properties 
• Use of the Wait procedure for visual effect and 

troubleshooting 
 

 
 
Design Notes 

 
This project further demonstrates how control arrays and variable arrays can be used together. 
 
Set the properties for one label, then copy and paste it.  When pasting, create a control array. 
 
It is very important to have the frame selected when pasting the shapes for the control arrays. Without selecting 
the frame, the pasted shape is not linked to it (if this occurs, simply delete the object, select the frame and 
paste the object again). 
 
Multiple controls can be selected by holding down SHIFT key as they are clicked. This will enable you to copy and 
paste the whole row together.  Always check the index numbers of each object after they have been copied.  
The order may not turn out as you would expect. 
 
Always set the name properties before entering any code.  Labels that serve no other purpose than displaying 
text do not need particular names.  The default names Label1, Label2 etc. are sufficient. 
 
 

Interface 
 
Create the interface as shown. 
Use 1 form, 2 separate frames each with a separate control array of 18 shapes, 4 command buttons and a 3rd 
frame with a control array of 4 option buttons.  
 
It is necessary to match the shape property of the shape with the index value of each option button.  Shape 
value 0 = rectangle; value 1 = square, value 2 = oval and value 3 = circle. 

Names of Objects 
 
Type of Object Number  Names of Objects Simple Initial Properties of Objects 

Form 1 Form1 Caption - “Variable Arrays 2” 
Icon - Browse and choose any icon 
Startup Position - Centre Screen 

Frames 3 No name necessary Captions - pattern1, pattern 2, shape style 
Font - Bold, 12 

Control arrays of 18 
shapes 

2 shpP1(0) to shpP1(17)  
shpP2(0) to shpP2(17) 

BackStyle - Opaque 
BackColor - White 
Shape - 0 (Rectangle) 

Control array of 
option buttons 

1 optStyle(0) to optStyle(3) Labels - as per interface 
OptStyle(1) Value - True 

Command Buttons 4 cmdGenerate, cmdStore, cmdRecall, cmdClear Font - Bold, 12  
Captions - as per interface 

 
 
Code 
 

 
GENERAL SECTION 
 
Dim z As Integer  'loop variable 
Const NumCells = 18 'Number of Shapes in Linear Pattern 
 
Dim Pattern1(NumCells - 1) As Long 
Dim ShapePattern1(NumCells - 1) As Integer 
Dim ShapeStyle As Integer 
 
Sub Wait(x As Single) 
 
Dim start As Single 
 
start = Timer 
While start + x > Timer 
    DoEvents 
Wend 
 
End Sub 
 
 
EVENTS 
 
Private Sub cmdGenerate_Click() 
 
Randomize 
For z = 0 To NumCells - 1 
    shpP1(z).BackColor = RGB(Rnd * 255, Rnd * 255, Rnd * 255) 
     
    shpP1(z).Shape = Int(Rnd() * 4)  'between 0 and 3 
    Wait 0.1    ‘for effect 
Next z 
 
End Sub 
 
 
Private Sub cmdRecall_Click() 
 
For z = 0 To NumCells - 1 
    shpP2(z).BackColor = Pattern1(z) 
    shpP2(z).Shape = ShapePattern1(z) 
Next z 
 
End Sub 

Private Sub cmdClear_Click() 
 
For z = 0 To NumCells - 1 
     
 shpP1(z).BackColor = vbWhite 
 shpP2(z).BackColor = vbWhite 
 shpP1(z).shape = 0 
 shpP2(z).shape  = 0 
      
Next z 
 
End Sub 
 
 
Private Sub cmdStore_Click() 
 
For z = 0 To NumCells - 1 
   Pattern1(z) = shpP1(z).BackColor 
 ShapePattern1(z) = shpP1(z).Shape 
 Next z 
 
End Sub 
 
 
Private Sub optStyle_Click(Index As Integer) 
 
ShapeStyle = Index 
 
For z = 0 To NumCells - 1 
    shpP1(z).Shape = ShapeStyle 
Next z 
 
End Sub 
 

 
 
 
Consolidation and Extension 

 
1. Which lines of code reset the shapes to rectangles? 

      
2. Write code that resets the shapes to circles. 

 
3. Name of the variable array that is storing the shape information about Pattern1? 

 
4. How could we randomly generate the Border Style and Border Width property of the patterns? 

 
5. What does each of the following lines do? 

 
   shpP1(z).BorderStyle = Int(Rnd() * 6) + 1 
 
   shpP1(z).BorderWidth = Int(Rnd() * 8) + 1 

  
 

6. Which section of code slows down the running of the program? 
  
7. ‘x as single’ is a parameter variable declaration.  How is this different to the other declarations? 
 
8. a. Modify the Frame containing the Shape Style option buttons so that it includes the rounded square and 

rounded rectangle. 
 

b. Modify the code accordingly to activate the new buttons. 
 

9. What other properties of the shapes could be randomly generated, stored and recalled? 
 

10. Name some other applications of control arrays. 
 

 


